Mouse Functions


Mouse Functions

MOU_ClipCursor() : Restricts the cursor to the specified area.

Remark

MOU_ClipCursor( .NULL. ) is identical to MOU_FreeClipCursor().MOU_ClipCursor() affects your whole Windows environment.

Alias

ClipCursor()

Syntax

MOU_ClipCursor( nTop,nLeft,nBottom,nRight ) lSuccess

Or…

MOU_ClipCursor( .NULL. ) lSuccess

Parameters

nTop
the top position of the rectangular area the cursor will be restricted to.

nLeft
the left position of the rectangular area the cursor will be restricted to.

nBottom
the bottom position of the rectangular area the cursor will be restricted to.

nRight
the right position of the rectangular area the cursor will be restricted to.

Or…

.NULL.
when the .NULL. value is passed to MOU_ClipCursor() the function behaves like MOU_FreeClipCursor(). When used so, the ClipCursor() alias is performing exactly as documented in the Win32 API.

The nTop, nLeft, nBottom and nRight parameters delimit a rectangular area described in absolute pixel screen coordinates. Therefore, should the developer want to restrict the coordinates to a particular region of a given form, then he'll need to adjust the relative position to absolute screen position. What's more, he should notice that he needs to pay attention to Window titles, menu height, form border width, form border height, etc. In the example below, the accurate calculations were performed.

Returns

.T.
the function always returns .T..

Example

oClipForm = CREATEOBJECT( "ClipForm" )
oClipForm.Show()

DEFINE CLASS clipForm AS form

    Name = "Dvlform1"

    ADD OBJECT dvlseparator1 AS dvlseparator WITH ;
        Top         = 202                       , ;
        Left        = 150                       , ;
        Width       = 210                       , ;
        Height      = 10                        , ;
        Name        = "Dvlseparator1"           , ;
        Line1.Name  = "Line1"                   , ;
        Line2.Name  = "Line2"                   , ;
        Line3.Name  = "Line3"                   , ;
        Line4.Name  = "Line4"

    ADD OBJECT dvlbutton1 AS dvlbutton WITH       ;
        Top         = 192                       , ;
        Left        = 361                       , ;
        Height      = 25                        , ;
        Width       = 10                        , ;
        Caption     = ""                        , ;
        Enabled     = .T.                       , ;
        Name        = "Dvlbutton1"

    ADD OBJECT dvllabel1 AS dvllabel WITH         ;
        Caption     = "0"                       , ;
        Height      = 15                        , ;
        Left        = 150                       , ;
        Top         = 220                       , ;
        Width       = 8                         , ;
        Name        = "Dvllabel1"

    ADD OBJECT shape1 AS shape WITH               ;
        Top         = 10                        , ;
        Left        = 150                       , ;
        Height      = 150                       , ;
        Width       = 150                       , ;
        FillStyle   = 4                         , ;
        BackColor   = RGB(255,255,0)            , ;
        Name        = "Shape1"

    ADD OBJECT shape2 AS shape WITH               ;
        Top         = 10                        , ;
        Left        = 310                       , ;
        Height      = 150                       , ;
        Width       = 52                        , ;
        FillStyle   = 4                         , ;
        BackColor   = RGB(255,0,0)              , ;
        FillColor   = RGB(255,255,255)          , ;
        BorderColor = RGB(0,0,0)                , ;
        Name        = "Shape2"

    *--------------------------------------------------------------*

    PROCEDURE dvlbutton1.MouseMove( nButton,nShift,nXCoord,nYCoord )

        LOCAL nPercent

        IF ( nButton == 1 )

            This.Left = nXCoord
            nPercent  = ( ( nXCoord - ThisForm.dvlSeparator1.Left ) / ;
                            ThisForm.dvlSeparator1.Width ) * 100

            ThisForm.dvlLabel1.caption = ALLTRIM(STR(nPercent))
            ThisForm.Shape1.Width      = 150 * ( nPercent / 100 )
            ThisForm.Shape1.Height     = 150 * ( nPercent / 100 )
            ThisForm.Shape2.Height     = ThisForm.Shape1.Height
            ThisForm.Shape2.Left       = ThisForm.Shape1.Left + ;
                                         ThisForm.Shape1.Width + 10
        ENDIF

    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE dvlbutton1.MouseUp( nButton,nShift,nXCoord,nYCoord )
        MOU_FreeClipCursor()
    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE dvlbutton1.MouseDown( nButton,nShift,nXCoord,nYCoord )

        LOCAL nTitleHeight
        LOCAL nBorderHeight
        LOCAL nMenuHeight
        LOCAL nBorderWidth

        nTitleHeight  = SYSMETRIC(9)
        nBorderHeight = SYSMETRIC(4)
        nMenuHeight   = SYSMETRIC(20)
        nBorderWidth  = SYSMETRIC(3)

        nTop  = _SCREEN.Top  + ThisForm.Top  + ThisForm.Dvlseparator1.Top + ;
                ( 2 * nTitleHeight ) + nMenuHeight + ( 2 * nBorderHeight )
        nLeft = _SCREEN.Left + ThisForm.Left + ;
                ThisForm.Dvlseparator1.Left + ( 2 * nBorderWidth )

        MOU_ClipCursor( nTop                                 , ;
                        nLeft                                , ;
                        nTop  + ThisForm.Dvlseparator1.Height, ;
                        nLeft + ThisForm.Dvlseparator1.Width ) && - This.Width)
    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE shape1.RightClick()
        MOU_FreeClipCursor()
    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE shape1.Click()

        LOCAL nTitleHeight
        LOCAL nBorderHeight
        LOCAL nMenuHeight
        LOCAL nBorderWidth

        nTitleHeight  = SYSMETRIC(9)
        nBorderHeight = SYSMETRIC(4)
        nMenuHeight   = SYSMETRIC(20)
        nBorderWidth  = SYSMETRIC(3)

        nTop  = _SCREEN.Top  + ThisForm.Top  + This.Top + ;
                ( 2 * nTitleHeight ) + nMenuHeight + ( 2 * nBorderHeight )
        nLeft = _SCREEN.Left + ThisForm.Left + This.Left + ( 2 * nBorderWidth )

        MOU_ClipCursor( nTop                                 , ;
                        nLeft                                , ;
                        nTop  + This.Height, ;
                        nLeft + This.Width ) && - This.Width)
    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE shape2.Click()

        LOCAL nTitleHeight
        LOCAL nBorderHeight
        LOCAL nMenuHeight
        LOCAL nBorderWidth

        nTitleHeight  = SYSMETRIC(9)
        nBorderHeight = SYSMETRIC(4)
        nMenuHeight   = SYSMETRIC(20)
        nBorderWidth  = SYSMETRIC(3)

        nTop  = _SCREEN.Top  + ThisForm.Top  + This.Top + ;
                ( 2 * nTitleHeight ) + nMenuHeight + ( 2 * nBorderHeight )
        nLeft = _SCREEN.Left + ThisForm.Left + This.Left + ( 2 * nBorderWidth )

        MOU_ClipCursor( nTop                                 , ;
                        nLeft                                , ;
                        nTop  + This.Height, ;
                        nLeft + This.Width ) && - This.Width)
    ENDPROC

    *--------------------------------------------------------------*

    PROCEDURE shape2.RightClick
        MOU_FreeClipCursor()
    ENDPROC

    *--------------------------------------------------------------*

ENDDEFINE

See also

MOU_FreeClipCursor()
MOU_FreeClipCursor() : Clears a previously set cursor clipping region.

Remark

MOU_FreeClipCursor() is identical to MOU_ClipCursor( .NULL. ).

Alias

FreeClipCursor() .T.

Syntax

MOU_FreeClipCursor() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

Example

See MOU_ClipCursor() example.

See also

MOU_ClipCursor()
MOU_isdown() : Is the left mouse button down?

Comment

Identical to MDOWN().

Syntax

MOU_isdown() lIsDown

Parameters

None.

Returns

lIsDown
.T. if the left mouse button is down

Example

Waiting until the left button is depressed
DO WHILE( MOU_isdown() )
ENDDO

MOU_LastVersion() : Returns the file stamp of MOU functions.

Remark

This function helps the developer identifying the last version of a set of functions. Sometimes the global version information of FOCUS.FLL (MIS_major() and MIS_minor()) does not help tracking down the changes in a project. Starting with version 6.0 of FOCUS.FLL, each source file has now an internal date and time stamp.

Syntax

MOU_LastVersion()  szLastVersion

Parameters

None.

Returns

szLastVersion
string identifying the last version of the functions set. The string is similar to "C:\Focus\5.0\MOUSE.C-Mon Oct 19 15:55:22 1998".

MOU_LeftDown() : Simulates a left button down.

Syntax

MOU_LeftDown() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

Example

* Simulates a click where the mouse is 
MOU_LeftDown()
MOU_LeftUp()

MOU_LeftUp() : Simulates a left button up.

Syntax

MOU_LeftUp() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

Example

* Simulates a click where the mouse is 
MOU_LeftDown()
MOU_LeftUp()

MOU_MiddleDown() : Simulates a middle button down.

Syntax

MOU_MiddleDown() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

MOU_MiddleUp() : Simulates a middle button up.

Syntax

MOU_MiddleUp() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

MOU_RightDown() : Simulates a right button down.

Syntax

MOU_RightDown() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

Example

* Simulates a right click where the mouse is 
MOU_RightDown()
MOU_RightUp()

MOU_RightUp() : Simulates a right button up.

Syntax

MOU_RightUp() .T.

Parameters

None.

Returns

.T.
the function always returns .T..

Example

* Simulates a right click where the mouse is 
MOU_RightDown()
MOU_RightUp()

MOU_Swap() : Swap mouse buttons.

Remarks

Button swapping is provided as a convenience to people who use the mouse with their left hands. The MOU_Swap() function is usually called by Control Panel only. Although an application is free to call the function, the mouse is a shared resource and reversing the meaning of its buttons affects all applications.

Alias

MOU_SwapButton(), MOU_SwapButtons(), MOU_SwapMouseButton(), MOU_SwapMouseButtons(), SwapMouseButton().

Syntax

MOU_Swap( lSwap ) lSuccess

Parameters

lSwap
.T. to swap mouse buttons; .F. to restore mouse buttons.

Returns

lSuccess
.T.. if the function was successful; .F. if not.

Example

IF ( MOU_Swap( .T. ) )
    ? "Brace yourself! Enjoy the mouse now!"
ENDIF







